


Før kalkulus

GeoGebra i R2



Sindre Sogge Heggen

Forord

Dette dokumentet tar for seg funksjonaliteter i GeoGebra som egner seg for å løse oppgaver i matematikk R2. De syv første seksjonene er parallelle med de syv kapitlene i boken *Før kalkulus; Teoridel*. Det samsvarende kapitlet fra teoridelen kan med fordel leses før man her gir seg i kast med en seksjon.

De unummurerte seksjonene inneholder funksjonaliteter som går på tvers av temaene i R2, og blir brukt i noen eksempler og løsningsforslag fra de nummererte seksjonene.

Obs! Det er ikke lagt vekt på de formelle kravene til levering av digitale oppgaver ved skriftlig eksamen. For dette, se løsningsforslag for Del 2 i [Eksempeloppgaven](#) fra Udir.

Alt innhold er laget av Sindre Sogge Heggen. Teksten er skrevet i \LaTeX og figurene er lagd vha. \LaTeX og GeoGebra.

01.08.2017

Innhold

1	Følger og rekker	3
2	Trigonometri	5
3	Vektorer i rommet	8
4	Romgeometrier	9
5	Derivasjon og funksjonsdrøfting	14
6	Integrasjon	15
7	Differensialligninger	16
	Grafikkfeltet	20
	CAS	21
	Knapper	24
	Hurtigtaster	25
	Kommandoliste	25

1 Følger og rekker

Regresjonsanalyse (Regneark)

Analyse av tallfølge skrevet inn i regnearket for å finne en eksplisitt formel.

Eksempel

Finn den eksplisitte formelen til følgen


2,6, 22, 56, 114, 202, ...

Svar:


Vi velger **Vis ► Regneark** og skriver tallene inn i en tabell med leddnummeret i første kolonne og verdien i andre.

▼ Regneark		
	f_x	F K
	A	B
1	1	2
2	2	6
3	3	22
4	4	56
5	5	114
6	6	202

Vi markerer så hele tabellen, i verktøyen som da dukker opp, velger vi **Regresjonsanalyse**.



På vinduet som da kommer velger vi **Analyser**, og trykker deretter på **Vis statistikk** (Σx). I analysevinduet søker vi nå å finne en **Regresjonsmodell** hvor vi får^a $R^2 = 1$ i statistikkvinduet. I dette tilfellet gir et tredjegradspolynom det vi ønsker:



Av $y = x^3 - 3x + 4$ i figuren over konkluderer vi med at den eksplisitte formelen til følgen er:

$$a_n = n^3 - 3n + 4$$

^a R^2 er et mål på hvor godt modellen samsvarer med inputen, gitt som en skala mellom 0 og 1. 1 betyr fullstendig samsvar.

Sum(<Uttrykk>, <Variabel>, <Start>, <Slutt>) (CAS)

Finner summen av en rekke med en løpende variabel på et intervall.

Eksempel

Finn summen av den uendelige rekke

$$1 + \frac{1}{5} + \frac{1}{25} + \dots$$

Svar:

Dette er en geometrisk rekke med $k = \frac{1}{5}$ og eksplisitt formel gitt som:

$$a_n = \frac{1}{5^{(n-1)}}$$

for $n \in \mathbb{N}$.

I CAS skriver vi da (∞ -tegnet finner du ved å trykke på α -tegnet oppe i høyre hjørne):


1	Sum[1/5^(n-1), n, 1, ∞]
○	→ $\frac{5}{4}$

2 Trigonometri

Løs(<Likning med x>) (CAS)

Løser en likning med x som ukjent.

Eksempel 1



I *Før Kalkulus*; *Teoridel* brukes $n \in \mathbb{Z}$ som heltallsvariabel, GeoGebra bruker en indeksert k (her $k_1 \in \mathbb{Z}$).


Merk: Du kan også løse ligningen ved å skrive den inn i en celle og deretter trykke på **Løs**.

Eksempel 2

Løs ligningen

$$\cos^2(3x) - 3 \cos(3x) - 4 = 0$$

Svar:



Merk: Løsningen kan komprimeres til (forklar for deg selv hvorfor):

$$x = \frac{1}{3} \pi (2k + 1)$$

for $k \in \mathbb{Z}$.

TrigKombiner(<Funksjon>, sin(x))

Skriver om en funksjon på formen $a \sin(kx) + b \cos(kx)$ til et kombinert uttrykk på formen $r \sin(kx + c)$.

Eksempel

CAS	
1	TrigKombiner[sqrt(3)sin(x)+cos(x), sin(x)]
<input type="radio"/>	$\rightarrow 2 \sin\left(x + \frac{1}{6} \pi\right)$

RegSin(<Liste>)

Bruker regresjon med en sinusfunksjon for å tilpasse punkt gitt i en liste.

Eksempel

Gitt tabellen


x	$f(x)$
0	0
1	-2.12
2	-2.73
3	-0.62
4	2.37
5	2.88

Bruk regresjon for å finne en tilnærming til $f(x)$ uttrykt som en sinusfunksjon.


Svar:

Vi velger **Vis** ► **Regneark** og skriver inn tabellen. Vi markerer så begge kolonner, høyreklikker innenfor markeringsfeltet og velger

Lag ► **Liste med punkt**:



Om vi ønsker at alle punktene skal vises i grafikkfeltet, høyreklikker vi på grafikken og velger **Vis alle objekt**. Deretter skriver vi `RegSin[Liste1]` i kommandolinjen, og får funksjonen $f(x)$ i algebrafeltet og grafen til f i grafikkfeltet. Denne funksjonen er en tilnærming til $f(x)$ gitt i oppgaven.



3 Vektorer i rommet

Punkt(<Liste>)

Lager et punkt med koordinater gitt som liste.

Merk: For å lage punktet (x, y, z) kan man liksågodt skrive $(\mathbf{x}, \mathbf{y}, \mathbf{z})$ i inntastingsfeltet. Skriver man $(\mathbf{x}, \mathbf{y}, \mathbf{z})$ i CAS lager man vektoren $[x, y, z]$.

Vektor(<Punkt>)

Lager vektoren fra origo til et gitt punkt.

Merk: I CAS kan man lage vektoren $[x, y, z]$ ved å skrive $(\mathbf{x}, \mathbf{y}, \mathbf{z})$, dette anbefales.

Eksempel

Gitt vektorene $\vec{u} = [-4, 2, 7]$, $\vec{v} = [4, 6 + s, -(s + t)]$ og $\vec{w} = [12, 2t - 9s, 3s - t]$.

- a) Finn s og t slik at $\vec{v} \parallel \vec{w}$.
b) Bestem s slik at $\vec{u} \perp \vec{v}$ når $t = -2$.

Svar:

a) Det er en litt spesiell sak i CAS at en vektor $[x, y, z]$ definert ved å skrive $(\mathbf{x}, \mathbf{y}, \mathbf{z})$ vil ha en bedre funksjonalitet enn hvis den defineres ved **Vektor**-kommandoen. Vi starter derfor med å definere \vec{v} og \vec{w} på følgende måte (se [Definere variabler](#)):

CAS	
	$v := (4, 6 + s, -(s + t))$
1	$\rightarrow \mathbf{v} := \begin{pmatrix} 4 \\ s + 6 \\ -s - t \end{pmatrix}$
	$w := (12, 2t - 9s, 3s - t)$
2	$\rightarrow \mathbf{w} := \begin{pmatrix} 12 \\ 2t - 9s \\ 3s - t \end{pmatrix}$

Vi utnytter videre at $\vec{v} \parallel \vec{w}$ hvis $r\vec{v} = \vec{w}$, for en konstant r . Vi skriver denne ligningen inn i CAS og trykker så på **Løs**:

3	$r * v = w$
<input type="radio"/>	Løs: $\{\{\mathbf{r} = 3, \mathbf{s} = -1, \mathbf{t} = 3\}\}$

Vi har altså at $s = -1$ og $t = 3$.

b) Skal $\vec{u} \perp \vec{v}$, må vi ha at $\vec{u} \cdot \vec{v} = 0$. Vi definerer \vec{u} og bruker **ByttUt**-

kommandoen for å sette $t = 2$ i uttrykket til \vec{v} . Med det endrede uttrykket løser vi ligningen for skalarproduktet (se kommandoen Skalarprodukt på s. ??). CAS fjerner $*$ når vi skriver $u\$5$).

4	$u := (-4, 2, 7)$
5	$\rightarrow u := \begin{pmatrix} -4 \\ 2 \\ 7 \end{pmatrix}$
5	ByttUt[v, t, 2]
6	$\rightarrow \begin{pmatrix} 4 \\ s + 6 \\ -s - 2 \end{pmatrix}$
6	$u\$5 = 0$
	Løs: $\left\{ s = -\frac{18}{5} \right\}$

Skalarprodukt(<Vektor>, <Vektor>)

Finner skalarproduktet av to vektorer.

Merk: For to vektorer u og v kan man like gjerne skrive $u*v$.

Vektorprodukt(<Vektor>, <Vektor>) (CAS)

Finner vektorproduktet av to vektorer. (Merk: For to vektorer u og v kan man like gjerne skrive $u \otimes v$. Hurtigtast for \otimes er **alt+shift+8**).

Vinkel(<Vektor>, <Vektor>)

Gir vinkelen mellom to vektorer. Kan også brukes for vinkel mellom plan/linjer, plan/plan og linje/linje

4 Romgeometrier

Pyramide(<Punkt>, <Punkt>, ...)

Framstiller en pyramide i Grafikkfelt 3D. `Pyramide[A,B,C,D]` lager en pyramide med grunnflate A, B, C og toppunkt D , mens `Pyramide[A,B,C,D, E]` har grunnflate A, B, C, D og toppunkt E . Under kategorien *Pyramide* i algebrafeltet finner man en konstant som oppgir volumet til pyramiden.

Høyde(<Objekt>)

Gir avstanden fra toppunkt til grunnflate i et objekt.


Merk: Avstanden har retning, og derfor kan den noen ganger være negativ. Tallverdien er den geometriske høyden.

Eksempel

Finn volumet og høyden til tetraedet med grunnflate gitt ved punktene $A = (3, 2, 1)$, $B = (6, 2, 1)$, $C = (3, 6, 1)$ og toppunkt $D = (5, 4, 3)$.

Svar:

Vi skriver inn punktene og bruker deretter kommandoen `Pyramide[A, B, C, D]` for å lage tetraedet a . Algebrafeltet gir oss da at volumet til a er 4. I celle 1 finner vi høyden, som er 2.



Prisme(<Punkt>, <Punkt>, ...)

Framstiller en prisme i Grafikkfelt 3D. `Prisme[A,B,C,D]` lager en prisme med grunnflate ABC og tak DEF , `Prisme[A,B,C,D,E]` har grunnflate $ABCD$ og tak EFG . F, G og eventuelt E blir konstruert av GeoGebra slik at hver sideflate er et parallelogram. Under kategorien *Prisme* i algebrafeltet finner man en konstant som oppgir volumet til pyramiden.

Kurve(<Uttrykk>, <Uttrykk>, <Uttrykk>,
<Parametervariabel>, <Start>, <Slutt>)

Viser parameteriseringen av en kurve i Grafikkfelt 3D på et gitt intervall. Uttrykkene er henholdsvis uttrykkene for x, y og z -koordinatene, bestemt av en gitt parametervariabel.

Merk: Med mindre et bestemt intervall av kurven er ønsket, er det bedre å skrive parameteriseringen direkte inn i inntastingsfeltet som $A+t*u$, hvor A er et punkt på linja og u er en retningsvektor.

Linje(<Punkt>, <Punkt>)


Gir uttrykket til en linje mellom to punkt. Hvis punktene har tre koordinater består uttrykket av et punkt på linja og en fri variabel λ multiplisert med en retningsvektor.

Eksempel

Finn parameteriseringen til linjen som går mellom punktet $(-2, 1, 3)$ og $(-5, -1, 4)$.

Svar:

Vi skriver punktene i inntastingsfeltet og får punktene A og B . Etterpå skriver vi $\text{Linje}[A, B]$ og får da linjen f .



Av dette finner vi at parameteriseringen til linjen er gitt som:

$$f : \begin{cases} x = -2 - 3t \\ y = 1 - 2t \\ z = 3 + t \end{cases}$$

Kule(<Punkt>, <Radius>)

Viser en kule i Grafikkfelt 3D med sentrum i et gitt punkt og med en gitt radius.

Eksempel

En linje l med parameteriseringen

$$l : \begin{cases} x = -3 - 2t \\ y = 1 + t \\ z = 4 + 2t \end{cases}$$

skjærer en kule med sentrum i $(-1, 2, 6)$ og radius lik 3.


- a) Tegn kulen og linjen.
- b) Finn skjæringspunktet mellom kulen og linjen.

Svar:

Vi skal her se på to løsningsmetoder. Den første metoden er helt klart den raskeste, men den andre metoden er tatt med for å illustrere bruken av *Kurve*-kommandoen, i tillegg til å presentere en metode som vil sikre oss eksaktverdier.


Løsningsmetode 1

a)



Vi starter med å tegne kulen. I inntastingsfeltet skriver vi `Kule[(-1, 2, 6), 3]` og får kulen a i algebrafelt og grafikkfelt 3D. For å tegne linjen, skriver vi `(-3, 1, 4)+t*(2,1,2)` i inntastingsfeltet, resultatet er kurven f .

b)




I inntastingsfeltet skriver vi `Skjæring[a, f]` og får de to punktene A og B .

Merk: Hadde vi tegnet linjen ved hjelp av `Kurve`-kommandoen, ville ikke dette funket. `Skjæring` er ikke kompatibel med `Kurve`, og i dette tilfellet heller ikke med `CAS`.

Løsningsmetode 2

a)



For å tegne linjen, skriver vi `Kurve[-3 + 2t, 1 + t, 4 + 2t, t, -10, 10]` i inntastingsfeltet. At $t \in [-10, 10]$ velger vi ut ifra inspeksjon i grafikkfelt 3D. Det gjelder å velge et intervall som viser begge skjæringspunktene mellom kulen og linjen (man kan velge $t \in [-\infty, \infty]$, men da blir ikke kurven vist grafikkfeltet). Resultatet er kurven b .

b) (Se [Høyre- og venstresiden](#))

CAS	
1	$k(x, y, z) := \text{VenstreSide}[a]$ $\rightarrow \mathbf{k(x, y, z) := (x + 1)^2 + (y - 2)^2 + (z - 6)^2}$
2	$k(-3+2t, 1+t, 4+2t)=9$ Løs: $\{t = 0, t = 2\}$
3	$b(0)$ $\rightarrow \mathbf{(-3, 1, 4)}$
4	$b(2)$ $\rightarrow \mathbf{(1, 3, 8)}$

I celle 1 lager vi oss en ny funksjon $k(x, y, z)$ med et uttrykk tilsvarende venstresiden til kuleligningen. For at linjen skal skjære kule, må parameteriseringen til linjen oppfylle kuleligningen. I celle 2 setter vi derfor uttrykkene for x, y og z fra parameteriseringen inn i k , og krever at dette uttrykket skal bli lik 3^2 . Vi trykker så på Løs-knappen og får to svar for t . I celle 3 og 4 finner vi punktene for disse valgene av t .

`Plan(<Punkt>, <Punkt>, <Punkt>)`

Viser et plan i Grafikkfelt 3D, utspent av to av vektorene mellom tre gitte punkt.

5 Derivasjon og funksjonsdrøfting

`Deriverte(<Funksjon>)`

Gir den deriverte av en funksjon.

Merk: For en definert funksjon $f(x)$, kan man like gjerne skrive $f'(x)$.

Eksempel

CAS	
1	$f(x) := x^2$ $\rightarrow \mathbf{f(x) := x^2}$
2	$\text{Derivert}[f]$ $\rightarrow \mathbf{2x}$
3	$f'(x)$ $\rightarrow \mathbf{2x}$

Vendepunkt(<Polynom>)

Finner vendepunktene til et polynom.

Eksempel

CAS	
1	$f(x) := 3x^4 - 2x^2$
<input checked="" type="radio"/>	$\rightarrow f(x) := 3x^4 - 2x^2$
2	Vendepunkt[f]
<input type="radio"/>	$\rightarrow \left\{ \left(-\frac{1}{3}, -\frac{5}{27} \right), \left(\frac{1}{3}, -\frac{5}{27} \right) \right\}$

Maks(<Funksjon>, <Start x-verdi>, <Slutt x-verdi>)

Finner absolutt maksimum og maksimumspunkt for en funksjon f på et gitt intervall.

Min(<Funksjon>, <Start x-verdi>, <Slutt x-verdi>)

Finner absolutt minimum og minimumspunkt for en funksjon f på et gitt intervall.

Ekstremalpunkt(<Funksjon>, <Start>, <Slutt>)

Finner alle lokale ekstremalpunkt og ekstremalverdier for en funksjon f på et gitt intervall.

6 Integrasjon

Integral(<Funksjon>)

Gir uttrykket til det ubestemte integralet av en funksjon. (Merk: Hvis kommandoen skrives i inntastingsfeltet, blir konstantleddet utelatt).

Eksempel

CAS	
1	Integral[x^2]
<input type="radio"/>	$\rightarrow \frac{1}{3}x^3 + c_1$

c_1 er en vilkårlig konstant.

`Integral(<Funksjon>, <Start>, <Slutt>)`

Gir det bestemte integralet av en funksjon på et intervall.

Eksempel 1

CAS	
1	Integral[x^2, 0, 2]
<input type="radio"/>	→ $\frac{8}{3}$

Eksempel 2

Finn volumet av omdreingslegemet til $f(x) = x^2$ på intervallet $[0, 1]$.

Svar:

CAS	
1	f(x):= x^2
<input type="radio"/>	→ $f(x) := x^2$
2	$\pi * \text{Integral}[f^2, 0, 1]$
<input type="radio"/>	→ $\frac{1}{5} \pi$

I celle 1 definerer vi $f(x)$. Volumet er gitt som $\pi \int_0^1 (f(x))^2 dx$, som vi finner i celle 2.

7 Differensialligninger

`LøsODE(<Likning>) (CAS)`

Finner generell løsning av en gitt differensialligning av første eller andre orden.

Eksempel 1

Løs ligningen:

$$y' + 2y = 2$$

Svar:

CAS	
1	LøsODE[y'+2y=2]
<input type="radio"/>	→ $y = c_1 e^{-2x} + 1$

c_1 er en vilkårlig konstant.

Eksempel 2

Løs ligningen:

$$y' + 5y^2 = 0$$

Svar:

CAS	
1	LøsODE[y'+5y^2=0]
<input type="radio"/>	$\rightarrow y = \frac{1}{c_1 + 5x}$

c_1 er en vilkårlig konstant.

Eksempel 3

Løs ligningen:

$$y'' + y' - 6y = 0$$

Svar:

CAS	
1	LøsODE[y''+y'-6y=0]
<input type="radio"/>	$\rightarrow y = c_1 e^{2x} + c_2 e^{-3x}$

c_1 og c_2 er vilkårlige konstanter.

LøsODE(<Likning>, <Punkt på f>, <Punkt på f'>) (CAS)

Finnes løsningen av en gitt differensialligning av første eller andre orden, for randverdier gitt som punkter.

Eksempel 1

Finne løsningen av ligningen

$$y' - 3y = 0$$

med randbetingelsen $y(0) = 5$.

Svar:

Randbetingelsen gir oss punktet $(x_0, y(x_0)) = (0, 5)$:

CAS	
1	LøsODE[y'-3y=0, (0,5)]
<input type="radio"/>	$\rightarrow y = 5 e^{3x}$

Eksempel 2

Løs ligningen:

$$y'' + y - 6 = 0 \quad , \quad y(0) = -1, y'(0) = 0$$

Svar:

Punktet på y er $(0, -1)$ og punktet på y' er $(0,0)$. Løsningen kan vi da finne via CAS:

CAS	
1	LøsODE[y''+y'-6y=0, (0,-1), (0,0)]
<input type="radio"/>	$\rightarrow y = -\frac{2}{5} e^{-3x} - \frac{3}{5} e^{2x}$

Retningsdiagram($f(x,y)$) (Inntastingsfelt)

Lager et retningsdiagram for en differensialligning hvor $f(x,y) = y'$.

Eksempel

Gitt differensialligningen

$$y' + xy = x$$


- a) Tegn et retningsdiagram for løsningene av ligningen.
- b) Tegn integralkurven for løsningen som krysser vertikalaksen når $y = 2$.

Svar:


- a) Vi starter med å finne y' :

$$y' = x - xy$$


I inntastingsfeltet skriver vi så `Retningsdiagram[x-x y]` og får dette bildet i grafikkfeltet (*Obs! x og y må skilles med mellomrom eller gangetegn*):



b) Vi starter med å løse ligningen for punktet $(0, 2)$:



Trykker vi på den hvite markøren (som blir blå) i celle 1, vil en funksjon bli definert og vist i grafikkfeltet:



Grafikkfeltet

Funksjoner på et gitt intervall


I mange oppgaver jobber vi med funksjoner som er definert på et gitt intervall. Når vi på eksamen skal levere grafen til en slik funksjon, er det krav om at vi tegner grafen bare på intervallet som er oppgitt.

La oss bruke funksjonen

$$A(x) = x^3 + 4x^2 + x - 2, \quad x \in [-4, 1]$$

som eksempel. For å tegne grafen til A på intervallet $-4 \leq x \leq 1$, bruker vi kommandoen `Funksjon(<Funksjon>, <Start>, <Slutt>)`. Vi skriver da:

```
A = Funksjon( (x-1)(x+2)(x+3)+4, -4, 1 )
```




Husk: Aksene justeres¹ ved enten å bruke **Flytt grafikkfelt** (), eller ved å holde inne **shift**-knappen på tastaturet mens man drar over aksene.

Nullpunkter og ekstremalpunkter på gitt intervall


Når vi må definere funksjoner på et bestemt intervall, blir vi stilt ovenfor et lite problem når vi skal finne nullpunkter og ekstremalpunkter. I figuren under har vi først definert funksjonen $f(x) = \sin(\pi x)$, og deretter g som f for $x \in [-3, 3]$. (`g = Funksjon(f, -3, 3)`).

¹Grafen bør strekke seg så godt over grafikkfeltet som mulig, men aller viktigst er at grafen er synlig på hele intervallet til definisjonsmengden.




Tenk nå at vi i GeoGebra ønsker å finne alle nullpunktene til g . For funksjoner som ikke består av polynomuttrykk, må vi bruke kommandoen `NullpunktIntervall(<Funksjon>, <Start>, <Slutt>)`. Vi skriver derfor:

```
NullpunktIntervall[ g, -3, 3]
```



Men g har åpenbart nullpunkt for $x \in \{-3, 3\}$ også, vi mangler altså to nullpunkt! Dette kommer av `NullpunktIntervall`-kommandoen bare finner *lokale* nullpunkt (se vedlegg E i teoridelen), hvis endepunktene er nullpunkt blir de altså ikke markert. Fordelen med å først definere¹ f og deretter g , er at om vi skriver `NullpunktIntervall[f, -3, 3]`, får vi det vi ønsker (de gamle punktene A-E er først slettet):



Problemet, og løsningen av det, vil være akkurat de samme når vi skal finne ekstremalpunkter (`Ekstremalpunkt(<Funksjon>, <Start>, <Slutt>)`)

¹ f er definert for alle $x \in \mathbb{R}$ og har dermed ingen endepunkt.

CAS

Definere variabler

Hvis vi ønsker å definere variabler som vi skal bruke i andre celler, må vi skrive $:=$. I figuren under er forskjellen mellom $=$ og $:=$ demonstrert med et forsøk på å finne $f'(x)$ til funksjonen $f(x) = x^2$:

CAS	
1	$f(x)=x^2$ $\rightarrow \mathbf{f(x) = x^2}$
2	$f'(x)$ $\rightarrow \mathbf{f'(x)}$
3	$f(x):=x^2$ $\rightarrow \mathbf{f(x) := x^2}$
4	$f'(x)$ $\rightarrow \mathbf{2 x}$

Av figuren legger vi også merke til at celle 3 og celle 4 er markert med en hvit runding. Dette indikerer at størrelsen vil vises i *Grafikkfelt* eller *Grafikkfelt 3D* hvis man trykker på markøren (den skal da bli blå).

Celle-referanser

Ofte kommer vi ut for situasjoner der vi ønsker å bruke uttrykket vi har funnet i tidligere celler. Som eksempel har vi i celle 1 skrevet inn volumet v av en kule med radius r , mens i celle 2 har vi volumet V av en kule med radius R . Ønsker vi å finne forholdet mellom disse, kan vi bruke cellereferanser som hjelpemiddel. For å referere til celle 1 skriver vi $\$1$ og for celle 2 skriver vi $\$2$. Forholdet $\frac{v}{V}$ kan vi da skrive som $\$1/\2 :

CAS	
1	$v = 4/3 \pi r^3$ $\rightarrow \mathbf{v = \frac{4}{3} r^3 \pi}$
2	$V = 4/3 \pi R^3$ $\rightarrow \mathbf{V = \frac{4}{3} R^3 \pi}$
3	$\$1 / \2 $\rightarrow \mathbf{\frac{v}{V} = \frac{r^3}{R^3}}$

Lister

Når et uttrykk står inni sløyfeparanteser $\{\}$, betyr det at det er laget en liste. En liste inneholder flere elementer som vi kan hente ut. Dette gjør vi ved å skrive paranteser bak listen, hvor vi angir nummeret til elementet i listen.

CAS	
1	$\{a, b, c\}$
<input type="radio"/>	$\rightarrow \{a, b, c\}$
2	$\$1(1)$
	$\rightarrow a$
3	$\$1(2)$
	$\rightarrow b$
4	$\$1(3)$
	$\rightarrow c$

Lister bruker vi også når vi skal løse ligninger med flere ukjente:

CAS	
1	$x+y+z=6$
<input type="radio"/>	$\rightarrow x+y+z=6$
2	$x+y=3$
<input type="radio"/>	$\rightarrow x+y=3$
3	$y+z=5$
<input type="radio"/>	$\rightarrow y+z=5$
4	$\text{Løs}[\{\$1, \$2, \$3\}]$
<input type="radio"/>	$\rightarrow \{\{x=1, y=2, z=3\}\}$

Høyre- og venstresiden

De fleste uttrykkene vi jobber med i CAS inneholder et $=$ tegn. Disse uttrykkene er en ligning med en venstre- og høyreside. Ofte ønsker vi å bruke uttrykket på bare én av disse sidene, og oftest høyresiden. Som eksempel har vi løst ligningen $(a+b)x = c$ og definert funksjonen $f(x) = dx^2$. Vi ønsker så å sette løsningen av ligningen inn i funksjonen. Dette gjør vi ved hjelp av `HøyreSide`-kommandoen (resultatet uten bruken av denne er vist i celle 4).

CAS	
1	Løs[(a+b)x=c] → $\left\{ \mathbf{x} = \frac{\mathbf{c}}{\mathbf{a} + \mathbf{b}} \right\}$
2	f(x):= d x^2 → $\mathbf{f(x) := d x^2}$
3	f(HøyreSide[\$1]) → $\left\{ \mathbf{d \left(\frac{c}{a + b} \right)^2} \right\}$
4	f(\$1) → $\left\{ \mathbf{d x^2 = d \left(\frac{c}{a + b} \right)^2} \right\}$

ByttUt

Noen ganger ønsker vi å endre en variabel i et uttrykk. For å gjøre dette kan vi anvende `ByttUt(<Uttrykk>, <Liste med forandringer>)`. La oss se på uttrykket

$$\frac{a+b}{c}$$

Vi ønsker nå å sette $a = d$, $b = 2$ og $c = f$. Dette kan vi gjør ved å skrive følgende:

CAS	
1	(a+b)/c → $\frac{\mathbf{a + b}}{\mathbf{c}}$
2	ByttUt[\$1,{a = d, b = 2, c = f}] → $\frac{\mathbf{d + 2}}{\mathbf{f}}$

En typisk oppgave i R2 kan være å løse en differensialligning av andre orden med initialbetingelser, og deretter bruke denne løsningen videre i oppgaven. Oppgaven handler ofte om noe som endrer seg med tiden. Når CAS løser differensialligninger, blir resultatet en funksjon $y(x)$. Vi kan da bruke `ByttUt`-kommandoen for å definere en ny funksjon $f(t)$.

CAS	
1	LøsODE[y''+y=0, (0, 1), (0, 2)] → $\mathbf{y = \cos(x) + 2 \sin(x)}$
2	f(t):=ByttUt[\$1,{x = t}] → $\mathbf{f(t) := 2 \sin(t) + \cos(t)}$

Knapper

Grafikkfelt

Knappene velges fra rullemenyer på verktøylinjen. Nummereringen av menyene er fra venstre.



Lager et nytt punkt. (Meny nr. 1)



Lager linje mellom to punkt. (Meny nr. 2)



Finner topp- og bunnpunkt til en funksjon. (Meny nr. 2)



Finner nullpunktene til en funksjon. (Meny nr. 2)



Finner skjæringspunkt mellom to objekt. (Meny nr. 3)



Lager vektoren mellom to punkt (Meny nr. 3)



Lager en tekstboks. (Meny nr. 10)



Flytter grafikkfeltet. Endrer verdiavstanden hvis man peker på aksene. (Meny nr. 10)

CAS



Gjengir uttrykket som er inntastet, ofte i forkortet form.



Gjengir uttrykket som er inntastet.



Gir tilnærmet verdi av et uttrykk (som desimaltall).



Gir eksaktløsningen av en ligning.



Gir tilnærmet løsning av en ligning som desimaltall.

Hurtigtaster

	Beskrivelse	PC	Mac
$\sqrt{\quad}$	kvadratroten	<code>alt+r</code>	<code>alt+r</code>
π	pi	<code>alt+p</code>	<code>alt+p</code>
∞	uendelig	<code>alt+u</code>	<code>alt+,</code>
\otimes	kryssprodukt	<code>alt+shift+8</code>	<code>ctrl+shift+8</code>
e	eulers tall	<code>alt+e</code>	<code>alt+e</code>
$^\circ$	gradtegn ($\frac{\pi}{180}$)	<code>alt+o</code>	<code>alt+o</code>

Kommandoliste

`abs(<x>)`

Finner lengden til et objekt x . (Merk: kan brukes til å finne lengden av en vektor).

Asymptote(<Funksjon>)

Finner asymptotene til en funksjon.

Avstand(<Punkt>, <Objekt>)

Gir avstanden fra et punkt til et objekt.

ByttUt(<Uttrykk>, <Liste med forandringer>) (CAS)

Viser et gitt uttrykk etter endring av variabler, gitt i en liste.

Deriverte(<Funksjon>)

Gir den deriverte av en funksjon.

Merk: For en definert funksjon $f(x)$, kan man like gjerne skrive $f'(x)$.

Ekstremalpunkt(<Funksjon>, <Start>, <Slutt>)

Finner alle lokale ekstremalpunkt og ekstremalverdier for en funksjon f på et gitt intervall.

Ekstremalpunkt(Polynom)

Finner alle ekstremalpunkt og ekstremalverdier til et polynom.

Funksjon(<Funksjon>, <Start>, <Slutt>)

Tegner en funksjon på et gitt intervall.

HøyreSide(<Likning>) (CAS)

Gir høyresiden til en likning.

HøyreSide(<Liste med likninger>) (CAS)

Gir en liste med høyresidene i en liste med ligninger.

Integral(<Funksjon>)

Gir uttrykket til det ubestemte integralet av en funksjon. (Merk: Hvis kommandoen skrives i inntastingsfeltet, blir konstantleddet utelatt).

Integral(<Funksjon>, <Start>, <Slutt>)

Gir det bestemte integralet av en funksjon på et intervall.

Integral(<Variabel>) (CAS)

Gir uttrykket til det ubestemte integralet til en funksjon av gitt variabel. (Brukes dersom man ønsker å integrere funksjoner avhengig av en annen variabel enn x).

Kurve(<Uttrykk>, <Uttrykk>, <Uttrykk>, <Parametervariabel>, <Start>, <Slutt>)

Viser parameteriseringen av en kurve i Grafikkfelt 3D på et gitt intervall. Uttrykkene er henholdsvis uttrykkene for x , y og z -koordinatene, bestemt av en gitt parametervariabel.

Merk: Med mindre et bestemt intervall av kurven er ønsket, er det bedre å skrive parameteriseringen direkte inn i inntastingsfeltet som $A+t*u$, hvor A er et punkt på linja og u er en retningsvektor.

Kule(<Punkt>, <Radius>)

Viser en kule i Grafikkfelt 3D med sentrum i et gitt punkt og med en gitt radius.

Løs(<Likning med x>) (CAS)

Løser en likning med x som ukjent.

Løs(<Liste med likninger>, <Liste med variabler>) (CAS)

Finner alle løsninger av en liste med ligninger med gitte variabel som ukjente.

Løs(<Likning>, <Variabel>) (CAS)

Finner alle løsninger av en gitt likning med en gitt variabel som ukjent.

LøsODE(<Likning>) (CAS)

Finner generell løsning av en gitt differensialligning av første eller andre orden.

LøsODE(<Likning>, <Punkt på f>, <Punkt på f'>) (CAS)

Finner løsningen av en gitt differensialligning av første eller andre orden, for randverdier gitt som punkter.

Maks(<Funksjon>, <Start x-verdi>, <Slutt x-verdi>)

Finner absolutt maksimum og maksimalpunkt for en funksjon f på et gitt intervall.

Min(<Funksjon>, <Start x-verdi>, <Slutt x-verdi>)

Finner absolutt minimum og minimumspunkt for en funksjon f på et gitt intervall.

Nullpunkt(<Polynom>)

Finner alle nullpunkter til et polynom.

NullpunktIntervall(<Funksjon>, <Start>, <Slutt>)

Finner alle nullpunkter på et gitt intervall til en hvilken som helst funksjon.

Plan(<Punkt>, <Punkt>, <Punkt>)

Viser et plan i Grafikkfelt 3D, utspent av to av vektorene mellom tre gitte punkt.

Prisme(<Punkt>, <Punkt>, ...)

Framstiller en prisme i Grafikkfelt 3D. $\text{Prisme}[A,B,C,D]$ lager en prisme med grunnflate ABC og tak DEF , $\text{Prisme}[A,B,C,D,E]$ har grunnflate $ABCD$ og tak EFG . F, G og eventuelt E blir konstruert av GeoGebra slik at hver sideflate er et parallelogram. Under kategorien *Prisme* i algebrafeltet finner man en konstant som oppgir volumet til pyramiden.

Punkt(<Liste>)

Lager et punkt med koordinater gitt som liste.

Merk: For å lage punktet (x, y, z) kan man liksågodt skrive $(\mathbf{x}, \mathbf{y}, \mathbf{z})$ i inntastingsfeltet. Skriver man $(\mathbf{x}, \mathbf{y}, \mathbf{z})$ i CAS lager man vektoren $[x, y, z]$.

Pyramide(<Punkt>, <Punkt>, ...)

Framstiller en pyramide i Grafikkfelt 3D. **Pyramide**[A,B,C,D] lager en pyramide med grunnflate A, B, C og toppunkt D , mens **Pyramide**[A,B,C,D, E] har grunnflate A, B, C, D og toppunkt E . Under kategorien *Pyramide* i algebrafeltet finner man en konstant som oppgir volumet til pyramiden.

RegLin(<Liste>)

Bruker regresjon med en rett linje for å tilpasse punkt gitt i en liste.

RegEksp(<Liste>)

Bruker regresjon med en eksponentialfunksjon for å tilpasse punkt gitt i en liste.

RegPoly(<Liste>, <Grad>)

Bruker regresjon med et polynom av gitt grad for å tilpasse punkt gitt i en liste.

RegPot(<Liste>)

Bruker regresjon med en potensfunksjon for å tilpasse punkt gitt i en liste.

RegSin(<Liste>)

Bruker regresjon med en sinusfunksjon for å tilpasse punkt gitt i en liste.

Retningsdiagram(f(x,y)) (Inntastingsfelt)

Lager et retningsdiagram for en differensialligning hvor $f(x, y) = y'$.

Skalarprodukt(<Vektor>, <Vektor>)

Finner skalarproduktet av to vektorer.

Merk: For to vektorer u og v kan man like gjerne skrive $u \cdot v$.

Skjæring(<Objekt>, <Objekt>)

Finner skjæringspunktene mellom to objekter.

Merk: Fungerer ikke for vektorer, og gir bare ett av punktene dersom funksjonene har flere skjæringspunkt.

Skjæring(<Funksjon>, <Funksjon>, <Start>, <Slutt>)

Finner skjæringspunktene mellom to funksjoner på et gitt intervall.

Sum(<Uttrykk>, <Variabel>, <Start>, <Slutt>) (CAS)

Finner summen av en rekke med en løpende variabel på et intervall.

TrigKombiner(<Funksjon>)

Skriver om et uttrykk på formen $a \sin(kx) + b \cos(kx)$ til et kombinert uttrykk på formen $r \cos(kx - c)$

TrigKombiner(<Funksjon>, sin(x))

Skriver om en funksjon på formen $a \sin(kx) + b \cos(kx)$ til et kombinert uttrykk på formen $r \sin(kx + c)$.

Vektor(<Punkt>)

Lager vektoren fra origo til et gitt punkt.

Merk: I CAS kan man lage vektoren $[x, y, z]$ ved å skrive **(x,y,z)**, dette anbefales.

Vektorprodukt(<Vektor>, <Vektor>) (CAS)

Finner vektorproduktet av to vektorer. (Merk: For to vektorer u og v kan man like gjerne skrive $u \otimes v$. Hurtigtast for \otimes er **alt+shift+8**).

Vendepunkt(<Polynom>)

Finner vendepunktene til et polynom.

VenstreSide(<Likning>) (CAS)

Gir venstresiden til en likning.

VenstreSide(<Liste med likninger>) (CAS)

Gir en liste med venstresidene i en liste med ligninger.

Vinkel(<Vektor>, <Vektor>)

Gir vinkelen mellom to vektorer. Kan også brukes for vinkel mellom plan/linjer, plan/plan og linje/linje